
	Система GPSS/PC
	каефедра  АСОИУ 


Лабораторная работа №4

 Системы с накопителями: блоки ENTER, LEAVE, оператор STORAGE.
Управление потоком сообщений: блоки GATE, TEST. Операторы VARIABLE, FVARIABLE, INITIAL.

Цель работы: Моделирование систем с конечной емкостью накопителей. Управление потоком сообщений на основе заданных соотношений и логических условий. Изучение опе​раторов VARIABLE, FVARIABLE, INITIAL.

В системах массового обслуживания часто задается ограниченная длина очереди или накопителя для требований, которые входят в систему, и ожидают своего обслуживания. Для моделирования систем с ограниченной очередью и параллельной обработкой требований в GPSS/PC используются блоки ENTER, LEAVE и оператор STORAGE. Эти блоки и оператор образуют группу памятей или группу описания накопителей, в которых собираются сообще​ния перед началом обслуживания в одном из устройств. В системе GPSS/PC памяти или на​копители относятся к аппаратной категории, которые имеют визуализацию в виде специаль​ных окон для наблюдения заполнения  накопителей (в рабочей строке GPSS/PC следует набрать сочетание клавиш Alt+S).
Блок GATE проверяет состояние устройств, памятей, логических ключей. Блок TEST сравнивает два стандартных числовых атрибута по вычисляемому отношению. Оба блока управляют потоком сообщений (транзактов).

Пример 1. Входной поток требований поступает в накопитель с допустимой емкостью, равной 5 единицам, равномерно в течение 4±1 мин. Если требования после накопителя за​стают 1-е устройство заполненным, то они поступают на обработку во 2-е устройство. Время обработки для 1-го устройства равно 13±1 мин., для второго — 11±1 мин. Смоделировать процесс обработки требований в течение 3 часов (180 мин.). 

 Решение примера 1 приводится как программа 1.

	Программа 1

	SIMULATE
nak STORAGE 5; Накопитель емкостью в 5 единиц
10  GENERATE 4,1

20  GATE SNF nak,FACIL2; Анализ состояния накопителя
30  ENTER nak
35  QUEUE 1

40  SEIZE 1; 1-е устройство
45  DEPART 1

50  LEAVE nak
60  ADVANCE 13,1

70  RELEASE 1
80  TERMINATE
100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2 

200  ADVANCE 11,1 

300  RELEASE 2

400   TERMINATE
***************************************************************************************

500  GENERATE 180

600  TERMINATE 1

START 1 

;END


В данном примере поступающие в систему требования сначала буферизуются в нако​пителе емкостью в 5 единиц, а потом пытаются попасть на обработку в 1-е устройство. Если первое устройство оказывается занятым, т.е. в нем осуществляется обслуживание, то очередное требование поступает на обслуживание во второе устройство. В программе используются следующие форматы записи оператора STORAGE, блоков  ENTER,  LEAVE и GATE.

Формат записи оператора STORAGE:

nak  STORAGE  5


Описание оператора находится вне основного тела программы, без нумерации. Метка nak служит для обращения или ссылки на оператор. Операнд поля <A> определяет объем накопителя (у нас 5). Операнд поля <A> должен быть положительным целым. Когда транзакт пытается войти в блок ENTER, то запрос на емкость накопителя сравнивается с доступной емкостью. Если запрос может быть выполнен, то транзакт входит в блок ENTER, и текущая емкость накопителя уменьшается. Если запрос транзакта не может быть удовлетворен, то транзакт входит в список задержки многоканального устройства. 
Формат записи блока ENTER:

30 ENTER  nak

В поле <A> указывается номер или имя (у нас nak) накопителя. Когда транзакт входит в блок ENTER, то операнд поля <A> используется для нахождения накопителя с указанным именем (унас накопитель с именем nak). Если такого накопителя не существует, то происходит ошибка выполнения программы.

Формат записи блока LEAVE:
50 LEAVE  nak

Блок LEAVE освобождает определенное число единиц накопителя под именем, заданным в поле <A> (у нас nak). Поле <A> также определяет номер или имя накопителя (у нас nak). Может быть использовано поле <B>, которое определяет  число освобождаемых единиц накопителя. Если это поле пусто, предполагается 1. Число освобождаемых единиц не должно превышать текущее содержимое накопителя. Обычно поле <B> не используется.

Формат записи блока GATE:

20 GATE  <X> <A>,<B>

Поле <A> содержит имя или номер объекта (в данном случае накопителя с меткой nak), для которого проводится проверка во  вспомогательном поле операции <X>. Поле <B> содержит номер или имя следующего блока для входящего транзакта, когда логический оператор поля <X> имеет значение "ложь". В поле <X> программы  задан логический оператор SNF, связанный с накопителем и означающий, что накопитель, заданный в поле <A>, заполнен. Пока накопитель будет не заполнен, т.е. логический оператор SNF будет давать "ложь", то тогда транзакт будет отправлен в блок, заданный в поле <B> — в блок с меткой FACIL2.

В противоположность условию SNF в поле <X> может быть задан логический оператор SF, который означает, что  накопитель не заполнен. 

Выходная статистика по устройствам и блокам поменяется местами, если применить оператор SF. Анализ работы системы начинается со второго устройства и если оно оказавается занято по обслуживанию текущего транзакта, то очередной транзакт будет отправлен на обработку в первое устройство. 

Задание к примеру 1. 

· Изучить и объяснить файл стандартного отчета для программы 1.
· Изменить среднее время поступления транзактов в систему в соответствии с данным рядом чисел: 2, 3, 5, 6, 7, 9, 13, 15, 21 (мин.).

· Изменить среднее время обработки первого устройства при среднем времени поступления требований в систему, равном 4 мин.: 7, 9, 11, 15. В то же время среднее время обработки второго устройства оставить неизменным. Изменить также среднее время обработки второго устройства при неизменном времени обработки первого устройства: 7, 9, 11, 15. Изучить и объяснить файл стандартного отчета. Изменить для блока GATE логическое условие с SNF на SF. Подобрать средние времена обработки первого и второго устройства так, чтобы выходная статистика по устройствам оставалась неизменной. 
Пример  блока GATE  с логическим оператором, связанным с устройствами.

Пример 2.  Входной поток требований поступает в систему с двумя обслуживающими устройствами равномерно  в течение 5(1 мин. Если  требования   застают первое устройство занятым, то они поступают на обработку во второе устройство. Время обработки для первого устройства равно 9(1 мин., для второго — 13(1 мин. Смоделировать процесс обработки требований в течение 3 часов (180 мин.).

Подобная ситуация характерна для систем, когда каждое входящее в систему требование в первую очередь отправляется на более быстродействующее устройство обслуживания. Если же это устройство занято, то приходится использовать устройство менее быстродействующее. Решение примера 2 приводится как программа 2.

	Программа 2


SIMULATE 

10  GENERATE 5,1
20  GATE NU 1,FACIL2
30  SEIZE 1; 1-е устройство

40  ADVANCE 9,1
50  RELEASE 1
60  TERMINATE 


600  TERMINATE 1

START 1
	END


Для этой программы рассмотрим формат записи блока GATE с логическим оператором — оператором NU, связанным с устройством:

20   GATE  NU  1,FACIL2
В поле <X> задан логический оператор NU, который указывает на то, что устройство, заданное в поле <A> (у нас устройство под номером 1), свободно. Если логическое условие не выполняется, то транзакт отправляется к блоку, имя которого задано в поле <B> (у нас отправление происходит по метке FACIL2). 

В противоположность оператору NU может быть использован оператор U — логический оператор, соответствующий занятому устройству.

Задание к примеру 2. 

· Изучить и объяснить файл стандартного отчета для программы 2.

· Ввести в программу блоки,  по которым создается статистика по очередям. 

· Изменить среднее время поступления транзактов в систему в соответствии с данным рядом чисел: 2, 3, 4, 6, 7, 9, 13, 15, 21 (мин.). 

· Изменить среднее время обработки первого устройства при среднем времени поступления требований в систему, равном 5 мин.: 7, 9, 11, 15. В то же время среднее время обработки второго устройства оставить неизменным. Изменить также среднее время обработки второго устройства при неизменном времени обработки первого устройства: 7, 9, 11, 15. Изучить и объяснить файл стандартного отчета.

· Изменить для блока GATE логическое условие с NU  на U. Для логического условия U  последовательно изменять параметры полей <A>, <B> блока GATE, получить загрузку обоих устройств или только одного из двух. Изучить и объяснить файл стандартного отчета.

Примеры по моделированию систем с использованием блока TEST. Применение операторов VARIABLE, INITIAL.

Пример 3.  Смоделировать процесс обслуживания равномерного потока требований с интервалом 5(1 мин двумя приборами: обслуживание на первом приборе длится 9(1 мин, на втором — 13(1 мин. Причем в течение первых 100 мин обслуживание производить на первом приборе, а по истечении 100 мин — на втором  приборе.

Пример 4.  Смоделировать процесс обслуживания равномерного потока требований с интервалом 5(1 мин. двумя приборами: обслуживание на первом приборе длится 9(1 мин., на втором — 13(1 мин. Причем в течение первых 100 мин. обслуживание производить на втором приборе, а по истечении 100 мин. — на первом приборе.

Условия примеров различаются в последовательности начального обслуживания требований, которое связано с процессом функционирования системы.

Программы решения примеров 3 и 4 представлены как программы 3 и 4.

	Программа 3

	SIMULATE 

ATR1  VARIABLE SQR(10000)

********************************************************

10  GENERATE 5,1

20  TEST LE C1,V$ATR1,FACIL2

30  SEIZE 1; 1-е устройство

40  ADVANCE 9,1

50  RELEASE 1

60  TRANSFER ,EXIT

100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2

200  ADVANCE 13,1

300  RELEASE 2

400 EXIT  TERMINATE 1

START 100

END


	Программа 4


SIMULATE 

	ATR1  VARIABLE SQR(10000)

********************************************************

10  GENERATE 5,1

20  TEST GE C1,V$ATR1,FACIL2

30  SEIZE 1; 1-е устройство

40  ADVANCE 9,1

50  RELEASE 1

60  TRANSFER ,EXIT

100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2

200  ADVANCE 13,1

300  RELEASE 2

400 EXIT  TERMINATE 1

START 100

END


В программах применены новые объекты GPSS/PC — переменные типа арифметических. Оператор описания арифметической переменной не входит в основное тело программы, он содержит следующие три поля:

1) поле метки, содержащее имя переменной (в программах — ATR1), которое используется при обращении к этой переменной; поле операции, содержит слово VARIABLE;

2) поле операндов, в котором задается выражение, используемое для вычисления значения переменной (в программах —  SQR(10000), что означает
[image: image1.wmf]10000

).
Формат описания арифметической переменной для программ 3,4:

ATR1  VARIABLE  SQR(10000)
Поле метки отделяется от поля операции (VARIABLE) как минимум одним пробелом, поле операндов (задаваемое выражение) отделяется от поля операции также пробелом. Пробелы между символами недопустимы. Первый пробел записи считается концом выражения. При любом обращении к переменной в любом месте программы должно употребляться выражение V$ATR1. Символ V обозначает тип объекта GPSS/PC —переменную. Знак $ употребляется для связи объекта как переменной с конкретным обозначением (например, ATR1).
Формат записи блока TEST:
20 TEST  <X>  <A>,<B>,<C>

Блок  TEST определяет номер следующего блока для вошедшего в него транзакта (сообщения) в зависимости от того, выполняется требуемое условие или нет. Блок управляет потоком транзактов, проверяя выполнение алгебраических отношений между значениями, заданных в полях <A> и <B>. Во вспомогательном поле операции <X> записывается один из шести условных операторов:

LE — меньше или равно (как в программе 3). Отношение истинное, если значение аргумента поля <A> меньше или равно значению аргумента поля <B>;

GE — больше или равно (как в программе 4). Отношение истинное, если значение аргумента поля <A> больше или равно значению аргумента поля <B>;
E — равно. Отношение истинное, если значения обоих аргументов полей <A> и <B> равны;

L — меньше. Отношение истинное, если значение аргумента поля <A> меньше значения аргумента поля <B>;
NE — не равно. Отношение истинное, если значения аргументов полей  <A> и <B> не равны.

Если отношение аргументов (числовых атрибутов), заданных в полях <A> и <B>, истинно, транзакт переходит к следующему по номеру блоку (который расположен  за блоком TEST). Если отношение ложно, транзакт переходит к блоку, номер или имя которого задан полем <C> (в программах по метке блока устройства FACIL2). В поле <C> задается номер или имя блока для входящего транзакта, если отношение величин, заданных в полях <A> и <B>, ложно. В программах 3 и 4 в поле <A> задан стандартный числовой атрибут С1 — условное относительное время моделирования. В поле <B> задана переменная через имя ATR1 (V$ATR1), которая численно равна 100.

В программе 3 пока значение условного относительного времени не достигнет 100, транзакты от блока TEST будут переходить к следующему по номеру блоку (к блоку SEIZE под номером 30). Как только величина С1 станет больше 100, транзакты будут переходить к блоку, имя которого определено в поле <C>, т.е. к блоку с меткой FACIL2 и с номером 100 — SEIZE. Пока С1(100, транзакты на обслуживание отправляются в 1-е устройство. Как только С1 станет больше 100, транзакты начинают отправляться на обслуживание в 2-е устройство.

В программе 4 транзакты после блока TEST будут направляться к следующему по номеру блоку — блоку SEIZE под номером 30, когда значение условного относительного времени С1 будет больше числа 100, которое определено через арифметическую переменную под именем ATR1. Если же значение С1 меньше или равно величины 100, то транзакты отправляются  по метке к блоку, заданный в поле <C>. Начало отсчета относительного времени моделирования начинается с нуля, поэтому сначала транзакты будут отправляться к блоку, заданному в поле <C>, т.е. к блоку SEIZE под номером 100.
В поле <В> блока TEST может быть задана ячейка сохраняемых величин, которая определяется с помощью оператора INITIAL. Использование оператора INITIAL дано в программе 5, которая решает ту же задачу, что и программа 3. 
	Программа 5

	SIMULATE 

initial x13,100; В ячейку под номером 13 записывается число 100

*********************************************************

10  GENERATE 5,1

20  TEST LE C1,x13,FACIL2; Сравнение С1 и ячейки х13

30  SEIZE 1; 1-е устройство

40  ADVANCE 9,1

50  RELEASE 1

60  TRANSFER ,EXIT

100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2

200  ADVANCE 13,1

300  RELEASE 2

400 EXIT  TERMINATE 1

START 100

END


Формат записи оператора INITIAL:
INITIAL  <A>,<B>

Оператор INITIAL записывается вне тела основной программы и может не нумераваться. Оператор INITIAL позволяет задавать начальные значения ячеек, элементов матриц и логических переключателей. 

В приводимой программе в поле <A> оператора  INITIAL задается ячейка под номером 13 (х13), в которую записывается число 100 (оно находится в поле <B>). Операнд может быть целым или именем. В программе ссылка на ячейку под заданным номером 13 производится как х13 (в поле <B> блока TEST). Фактически в поле <B> блока TEST задано число 100.
Задание к примерам 3, 4: 

· В программе 3 ввести переменные, по которым задать средние времена обработки — натуральные числа от 3 до 19 для первого устройства и от 19 до 3 для второго устройства, соответственно. В исходных программах поменять условия сравнения в поле <X> блока TEST на "меньше", "не равно". При этом произвести замену среднего времени обработки в первом устройстве — задать числа от 3 до 15. В программах 3, 4, 5 ввести блоки для регистрации данных об очередях перед устройствами.

· С помощью блоков TEST,  GATE написать программу по обработке требований в системе, состоящей из пяти устройств, и предусмотреть обработку либо только в четных по номеру устройствах, либо в нечетных. Длительность функционирования системы задать в течение 3 часов, длительность обработки в каждом из устройств 9(1 мин., интервал генерации требований 7(1 мин. Изучить и объяснить получаемые результаты моделирования.

Пример использования арифметических переменных

Арифметические переменные (переменные с фиксированной точкой) и арифметические переменные с плавающей точкой описываются соответственно операторами VARIABLE  и  FVARIABLE. Только при описании переменных с плавающей точкой (FVARIABLE) допускается применение дробных констант, округление результатов происходит после вычисления заданного выражения.
Пример 5. Вычислить и сохранить следующие величины:

Пример решения примера приводится в программе 6.

	Программа 6

	simulate

asl  fvariable (sqr(49)+1/2+3/2)

as2  variable (sqr(49)+1/2+3/2)

as3  fvariable (log(15)+sqr(8))

as4  variable (log(15)+sqr(8))

10  generate 5

20  savevalue mmml,v$asl

30  savevalue mmm2,v$as2

40  savevalue mmm3,v$as3

50  savevalue mmm4,v$as4

100  terminate 1

start 1

end


Задание к примеру 5. 

· Вычисленные значения с помощью операторов VARIABLE и FVARIABLE проверить на стандартном калькуляторе, сравнить и объяснить полученные результаты (на калькуляторе и по файлу стандартного отчета).

Лабораторная работа №5
Недоступность устройств. Блоки FUNAVAIL, FAVAIL.

Недоступность многоканальных устройств. Блоки SUNAVAIL, SAVAIL.
Накопление и синхронизация сообщений. Блоки GATHER, MATCH.

Логические ключи. Блоки LOGIC, GATE.
Цель работы: изучение возможностей системы GPSS/PC  для моделирования ситуаций с помощью блоков FUNAVAIL и FAVAIL,  SUNAVAIL и SAVAIL, когда в течение определенного времени устройство (многоканальное устройство) обслуживания становится недоступным.  Изучение блока GATHER для сбора сообщений одного семейства (ансамбля). Осуществление синхронизации продвижения двух транзактов одного ансамбля (семейства), движущихся по разным путям модели. Изучение ситуаций с коммутируемыми объектами с помощью блоков LOGIC и GATE.

Изучение состояний  недоступности  устройств  обслуживания: 

блоки FUNAVAIL, FAVAIL

Блоки FUNAVAIL, FAVAIL предусмотрены для адекватного отображения состояния недоступности в статистике использования устройства обслуживания требований, заявок и т.п. Блок FUNAVAIL делает недоступным устройство до тех пор, пока не вступит в действие блок FAVAIL. Эти блоки действуют совместно. Для блока FUNAVAIL предусмотрены специальные режимы работы, задание которых приводит к следующим действиям над транзактами (сообщениями и т.п.), связанными с недоступным устройством:

— управляющему транзакту, который занимал и генерировал прерывание на устройстве в тот момент, когда оно стало недоступным, может быть разрешено продолжение использования этого устройства;

— прерванные транзакты, которые были прерваны другими транзактами при использовании этого устройства, могут продолжать претендовать на устройство в течение периода недоступности; могут быть сняты из рассмотрения на использование устройства и направлены к вспомогательному блоку или могут претендовать на устройство только по окончании периода недоступности;

— над задержанными транзактами, которые ожидали устройство для его занятия или прерывания в момент, когда оно стало недоступным, совершаются те же действия, что и над прерванными транзактами.

Пример 6. Поток требований, поступающий в систему каждые 5 мин., сначала направляется в первое устройство обслуживания. Каждые 55 мин. на время 15 мин. первое устройство становится недоступным. При этом каждые  11 мин. на время 9 мин. делается попытка прервать поток требований, поступающих в первое устройство. В случае недоступности первого устройства требования направляются на обслуживание во второе, третье и четвертое  устройства.  Длительность обслуживания: в первом устройстве составляет 4 мин., в остальных — 11 мин. Смоделировать работу системы по обслуживанию 100 требований.

Решение примера 6 приводится как программа 6.

	Программа 6

	SIMULATE 

10  GENERATE 5; Генерирование потока требований

20 FACIL1  SEIZE 1; Устройство №1 с меткой FACIL1

30  ADVANCE 4

40  RELEASE 1

100 exit1  TERMINATE 1

**************************************************************

200  GENERATE 55; Генерирование потока команд по недоступности

210  FUNAVAIL 1,re,FACIL2,,re,FACIL3,re,FACIL4; Недоступность 1-го 

220  ADVANCE 15

230  FAVAIL 1; Снятие недоступности 1-го устройства
250  TERMINATE

**************************************************************

300 FACIL2  SEIZE 2; Устройство №2 с меткой  FACIL2

310  ADVANCE 11

320  RELEASE 2

350  TRANSFER ,exit1

**************************************************************

400 FACIL3  SEIZE 3; Устройство №3 с меткой  FACIL3
410  ADVANCE 11

420  RELEASE 3

450  TRANSFER ,exit1

**************************************************************

500 FACIL4  SEIZE 4; Устройство №4 с меткой  FACIL4

510  ADVANCE 11

520  RELEASE 4

550  TRANSFER ,exit1

**************************************************************

600  GENERATE 11; Генерирование потока прерываний

610  PREEMPT 1

620  ADVANCE 9

630  RETURN 1

650  TERMINATE

START 100; Задание числа счетчика завершений

;end


Блоки, связанные с устройствами, FUNAVAIL и FAVAIL, PREEMPT и RETURN  действуют попарно. Попытка прерывания первого устройства блоком  PREEMPT позволяет задействовать все рабочие поля блока FUNAVAIL. Один блок FUNAVAIL осуществляет распределение требований по устройствам.

Формат записи блока FUNAVAIL:

210   FUNAVAIL  1,re,FACIL2, ,re,FACIL3,re,FACIL4

В поле <A> задается имя или номер устройства (в программе номер 1), которое должно стать недоступным. В поле <B> задаются режимы работы с транзактами (требованиями, сообщениями), использующими устройство в момент, когда оно становится недоступным:

· режим RE (режим удаления — remove) означает, что занимавший устройство транзакт, больше не может претендовать на устройство. Удаленный транзакт попытается войти в альтернативный блок, заданный в поле <C>. В программе использован режим RE (re в поле <B>);
· режим СО (режим продолжения — continue) позволяет транзакту использовать устройство в период недоступности;
· если режим не задан (режим по умолчанию). то это означает, что транзакт, использующий устройство, прерывается до тех пор, пока устройство вновь не станет доступным.
В поле <C> задается номер или имя блока, к которому направляется транзакт, использующий устройство в момент, когда оно становится недоступным (в программе транзакт отправляется в блок с меткой FACIL2).

В поле <D> задается номер параметра управляющего транзакта (в программе не задан). Информация по полю <D> может быть использована для вычисления времени, которое управляющий транзакт должен провести в списке будущих событий, когда устройство становится недоступным.

Поле <E> может быть использовано в следующих допустимых режимах:

· режим RE (режим удаления) указывает на то, что прерванные транзакты (например, блоком PREEMPT) больше не могут претендовать на устройство. Эти транзакты пытаются войти в альтернативный блок, заданный в поле <F>. В программе  задан режим удаления (в поле <E> задан операнд "re");

· режим СО (режим продолжения) допускает использование устройства всеми транзактами во время периода недоступности. Транзакты занимают устройство в порядке их приоритета;

· если режим не задан (режим по умолчанию — пустое поле), то прерванный транзакт покидает список прерывания устройства, и не допускается использование устройства во время периода недоступности.

 В поле <F> задается номер или имя альтернативного блока, к которому направляется прерванный транзакт. Поле <F> используется, если в поле <E> задан режим удаления. В программе в поле <F> задан блок с меткой FACIL3.

Когда устройство становится недоступным поле <G> определяет действия с транзактами, пытающимися занять устройство или сгенерировать на нем прерывание (например, блоком PREEMPT) в момент, когда это устройство становится недоступным. Допускаются следующие режимы поля <G>:

· режим RE (режим удаления) указывает на то, что задержанные транзакты больше не должны претендовать на устройство. Эти транзакты не должны пытаться войти в блоки RETURN или RELEASE. Они удаляются из списков задержки с тем, чтобы в дальнейшем попытаться войти в альтернативный блок, заданный полем <H>. В программе используется режим удаления (в поле <G> задан операнд "re");

· режим СО (режим продолжения) предполагает, что всем транзактам будет разрешено использовать устройство во время периода его недоступности. Транзакты занимают устройство в порядке их приоритета;

· если режим не задан (режим по умолчанию), то это означает, что задержанные транзакты покидают список задержки или список ожидания устройства и не могут больше претендовать на устройство во время периода недоступности.

Поле <H> задает номер или имя блока, к которому направляются задержанные транзакты в режиме RE поля <G>. В программе в поле <H> задан блок с меткой FACIL4. 

Блок PREEMPT используется в простом режиме прерывания.
Распределение  транзактов по устройствам может быть изменено, если для одного из устройств  применить блок TERMINATE с пустым полем <A>. 

Формат записи блока  FAVAIL:

230  FAVAIL 1

В поле <A> блока  FAVAIL задается номер или имя устройство, с которого снимается недоступность. В программе в поле <A> задано число 1 — номер первого устройства. Блок  FAVAIL делает доступным заданное устройство после периода недоступности. Чаще блок FUNAVAIL используется только для осуществления недоступности устройства.

Задание к примеру 6:

· Произвести в программе  изменения по недоступности устройства: сделать недоступным 2-е устройство, затем 3-е, затем 4-е. Соответственно изменить перераспределение транзактов блоком FUNAVAIL. При каждом новом распределении транзактов задавать следующие интервалы времени поступления транзактов в систему: 7, 9, 11, 15, 17, 19 мин. Проанализировать и объяснить файл стандартного отчета по каждой новой программе.

Моделирование многоканальных устройств в GPSS/PC

Для моделирования параллельной работы устройств в системе GPSS/PC  вводится понятие многоканального устройства. Такое устройство характеризуется определенной емкостью накопителя поступающей информации на обработку. Емкость накопителя описывается оператором STORAGE. Кроме того, в системе GPSS/PC  существуют блоки для анализа, использования и создания недоступности многоканального устройства. К ним относятся  блоки использования накопителя ENTER и LEAVE; доступность и недоступность многоканального устройства осуществляется блоками SUNAVAIL и  SAVAIL.
Блок SUNAVAIL переводит многоканальное устройство (а по существу накопитель) в состояние недоступности. Блок SAVAIL переводит заданное  многоканальное устройство из состояния недоступности в состояние доступности.
Пример 7. Пусть входной поток требований поступает в накопитель с допустимой емкостью, равной 3 единицам, равномерно  в течение 5(1 мин. Если  поступающие требования застают накопитель в состоянии недоступности, то они направляются в 1-е устройство, в противном случае поступают на обработку во 2-е устройство. Время обработки для 1-го устройства равно 13(1 мин., для второго — 9(1 мин. Смоделировать процесс обработки 100 требований, если состояние недоступности накопителя происходит через каждые 29(1 мин. в течение 17 мин.

Решение примера 7 приводится как программа 7.

	Программа 7

	SIMULATE 

nak STORAGE 3; Накопитель емкостью в 3 единицы

*************************************************************

10  GENERATE 5,1,,100

20  GATE SNV nak,FACIL2; Анализ доступности накопителя 

30  ENTER nak

40 FACIL1  SEIZE 1; 1-е устройство с меткой FACIL1

50  LEAVE nak

60  ADVANCE 13,1

70  RELEASE 1

80  TRANSFER ,EXIT

100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2

200  ADVANCE 9,1

300  RELEASE 2

400 EXIT  TERMINATE 1

500  GENERATE 29,1

510  SUNAVAIL NAK

520  ADVANCE 17

530  SAVAIL NAK

550  TERMINATE

START 100

;END


В данном примере поступающие в систему требования сначала буферизуются в накопителе емкостью в 3 единицы. Потом, в зависимости доступности или недоступности накопителя, они  направляются в одно из устройств с помощью блока GATE.

Формат записи блока SUNAVAIL:

510  SUNAVAIL  NAK

В поле <A> блока задается имя накопителя (оператора STORAGE). В программе накопитель под именем NAK становится недоступным.

Формат записи блока SAVAIL:

530  SAVAIL  NAK

Блок SAVAIL переводит накопитель под именем NAK из состояния недоступности в состояние доступности. В поле <A> блока SAVAIL задается имя накопителя, которое переводится в состояние доступности. 

Блоки SUNAVAIL и SAVAIL работают совместно.

В данной программе перевод из состояния недоступности в состояние доступности происходит через 17 мин. (в поле <A> блока ADVANCE под номером 520 задано время задержки, равное 17 мин.). В то же время состояние недоступности наступает каждые 29(1 мин. Команды на недоступность генерируются блоком GENERATE под номером 500.

Формат записи блока GATE  в режиме SNV:

20  GATE  SNV  nak,FACIL2
В операционном поле <X> блока GATE задан логический операнд SNV — многоканальное устройство (накопитель) не доступно. В поле <A> определено имя накопителя (в программе nak), состояние которого проверяется. В поле <B> задается альтернативный блок (в программе FACIL2), если логический операнд поля <X> имеет значение "ложь". Если альтернативный блок задан, то блок GATE работает в режиме безусловного входа, т.е. транзакт всегда может войти в блок GATE. При этом если логический операнд имеет значение "истина", то транзакт пытается войти в следующий по номеру блок (в программе  блок под номером 30). Если же операнд имеет значение "ложь", то транзакт пытается войти в альтернативный блок, заданный полем <B> (в программе блок под номером 100 и с меткой FACIL2).

Перераспределение требований по устройствам может быть изменено блоком GATE с другим условием в операционном поле <X>.

Приведем программу под номером 8 для решения примера 7, в котором зададим условие доступности накопителя. Для этого случая в поле <X> блока GATE установим  логический операнд SV. Изменение режима работы блока GATE отразится на статистике устройств и накопителя.

	Программа 8

	SIMULATE 

nak STORAGE 3; Накопитель емкостью в 3 единицы

***************************************************************

10  GENERATE 5,1,,100

20  GATE SV nak,FACIL2; Анализ доступности накопителя 

30  ENTER nak

40 FACIL1  SEIZE 1; 1-е устройство с меткой FACIL1

50  LEAVE nak

60  ADVANCE 13,1

70  RELEASE 1

80  TRANSFER ,EXIT

100 FACIL2  SEIZE 2; 2-е устройство с меткой FACIL2

200  ADVANCE 9,1

300  RELEASE 2

400 EXIT  TERMINATE 1

500  GENERATE 29,1; Команды недоступности

510  SUNAVAIL NAK

520  ADVANCE 17

530  SAVAIL NAK

550  TERMINATE

START 100

;END




Задание к программам 7 и 8. 

· Видоизменить программы так, чтобы моделирование осуществлялось в течение 5 часов.

· В исходных программах изменить емкость накопителя: 2, 4, 5, 6, 7, 8, 9, 10. 

· В исходных программах команды недоступности генерировать по средним значениям из интервала от 15 до 50 мин.

· Запрограммировать ситуацию: если 1-е устройство не доступно, то транзакты должны покидать систему. Система функционирует 5 часов.

· Изучить и объяснить файлы стандартных отчетов.

Моделирование накопления транзактов,  синхронизации движения транзактов по модели, управление потоком транзактов в зависимости 

от состояния логического ключа
Блок GATHER осуществляет накопление сообщений (транзактов), принадлежащих одному и тому же набору (ансамблю), с последующей попыткой одновременной передачи их к следующему блоку.  Действие блока GATHER  аналогично действию блока ASSEMBLE. Отличие состоит в том, что после накопления в блоке числа транзактов, указанного в поле <A>, они все передаются в следующий блок. Блок GATHER  не уничтожает  вошедших транзактов.

Блок MATCH предназначен для синхронизации продвижения двух транзактов одного ансамбля (семейства), движущихся по разным путям модели. Для синхронизации необходимы два блока MATCH, находящиеся в соответствующих местах моделирующей программы и называемые сопряженными. 

Для представления в модели коммутируемых объектов используют логические ключи, моделируемые блоками LOGIC и GATE. Блок LOGIC устанавливает положение ключа, которое может быть проверено любым сообщением (транзактом) в любой части модели. Блок LOGIC используется для "включения", "выключения" или инвертирования состояния ключа. Блок GATE, в зависимости от состояния логического ключа, осуществляет управление потоком транзактов.

Пример 8. Смоделировать следующий  процесс прохождения 100 пакетов на почте.

 На участок сборки пакетов  поступают коробки, и письма с интервалом времени 25(4 единицы. На контроль коробки  затрачивается 4(1 единицы времени; контроль писем  производится последовательно со временем 2(1 единица времени на письмо. Операция сборки требует одновременного поступления коробки  и всех писем  и производится со временем 4(2 единицы. В одной коробке  должно быть 8 писем. Все процессы подчиняются равномерному закону.

Единица времени задается пользователем. Система GPSS/PC осуществляет моделирование в относительных единицах, поэтому в файле стандартного отчета указываемые единицы времени можно рассматривать как те, которые задаются пользователем (минуты, секунды и т.д.). Решение примера 8 приводится как программа 9.

	Программа 9


simulate

10  generate 25,4,,100

20  split 8,chan2

30 chan1  seize 1

40  advance 4,1

50  release 1

60  transfer ,comm

70 chan2  seize 2

80  advance 2,1

90  release 2

100  gather 8

	110 comm  assemble 9  

120 chan3  seize 3

130  advance 4,2

140  release 3

150 exit  terminate 1

start 100

;end


Формат записи блока GATHER:

100 gather  <A>

В поле <A> задается число транзактов, принадлежащих к одному семейству, которое надо накопить (в программе 8 транзактов). Операнд поля <A> может быть именем, положительным целым (как в программе число 8), СЧА, СЧА с параметром (СЧА$<параметр>). При входе транзакта в блок GATHER, интерпретатор GPSS/PC просматривает семейство, к которому он принадлежит, и проверяет, находится ли в данном блоке GATHER другой транзакт из того же семейства. Если в блок GATHER не вошло заданное число транзактов, транзакты, находящиеся в блоке, остаются в состоянии синхронизации. 

В программе после поступления в блок GATHER 8 транзактов  они все выходят из блока и поступают в следующий по номеру блок — блок под номером 110 ASSEMBLE. Блок ASSEMBLE по своему функциональному назначению собирает 9 транзактов и отправляет в следующий по номеру блок один транзакт. С помощью блока ASSEMBLE моделируется процесс сборки пакета — 8 писем и коробка.

Изучение блока MATCH.

Пример 9. Информация поступает на узел связи с интервалом времени 300(50 единиц. Обработку производят двое операторов. После завершения операции обработки, выполняемой первым оператором со временем 90 единиц и вторым со временем 60 единиц, производится операция сверки, время выполнения которой принимается равной нулю.  Затем третий оператор  производит формирование пакета информации со временем 40 единиц. Формирование начинается только после поступления всех отдельных частей информации (условных частей информации). Смоделировать процесс прохождения 100  пакетов информации. Программа решения примера 9 представлена как программа 10.

	Программа 10


simulate

10  generate 300,50

20  split 1,chan2

30 chan1  seize 1

40  advance 90

50 sss1  match sss2  

	60  release 1

70  transfer ,comm

80 chan2  seize 2

90  advance 60

100 sss2  match sss1

110  release 2

120 comm  assemble 2

130  seize 3

140  advance 40

150  release 3

200  terminate 1

start 100

;end


Из условия примера видно, что специальных мер по реализации прохождения точно 100 пакетов не требуется, так как время поступления условных частей информации на обработку заметно больше суммарного времени обработки: 300(50 против 90+60+40=190. В блоке  GENERATE не используется поле <D> с установкой числа 100. Синхронизация прохождения транзактов по модели осуществляется с помощью двух блоков MATCH. Один блок MATCH установлен как бы в первом устройстве, а другой — во втором. Транзакты не отправляются в блок ASSEMBLE (на формирование пакета) до тех пор, пока оба устройства (оба оператора) не закончат требуемую обработку. 

Формат записи блоков MATCH:

50 sss1    match  <A>

100 sss2  match  <A>

В поле <A> блока MATCH (под номером 50 и меткой sss1) в программе задается имя сопряженного блока MATCH (метка sss2 под номером 100). Синхронизация движения транзактов  осуществляется путем подбора пар транзактов из одного ансамбля и задержки этих транзактов до тех пор, пока оба транзакта из одной пары не поступят в заданные точки модели (в программе  это выходы устройств — блоки RELEASE под номерами 60 и 110, откуда они поступают далее на формирование пакета  в блок ASSEMBLE). 

Без условий синхронизации пример 9 можно видоизменить к следующему виду.

Пример 10. Информация поступает на узел связи с интервалом времени 300(50 единиц. Обработку производят двое операторов. После завершения операции обработки, выполняемой первым оператором со временем 90 единиц и вторым со временем 60 единиц, информация отправляется к третьему оператору. Третий оператор  производит формирование пакета из поступивших частей информации со временем 40 единиц. Смоделировать процесс прохождения 100 пакетов.

Программа решения примера 10 представлена как программа 11.

	Программа 11

	simulate

10  generate 300,50

20  split 1,chan2

30 chan1  seize 1

40  advance 90

50  release 1

60  transfer ,comm

70 chan2  seize 2

80  advance 60

90  release 2

100 comm  assemble 2

110  seize 3

120  advance 40

130  release 3

200  terminate 1

start 100

;end




Программа 11 по сравнению с программой 10 имеет на два блока меньше, но это приводит к иным результатам. Анализ результатов произвести по файлам стандартного отчета. 

Работа логического ключа

Пример 11. На обработку поступают 101 требование с интервалом времени 7(1 единицы. Нечетные по номеру требования обрабатываются на первом приборе со временем 15 единиц, четные — на втором приборе со временем 17 единиц. Смоделировать процесс прохождения и обработки требований в системе.

Формирование потока четных и нечетных по номеру требований осуществляется блоками LOGIC и GATE (в режиме LS).

Решение примера 11 приводится в программе 12. 

	Программа 12


simulate

10  generate 10,1,,101

20  logic I key1; Задание операции инвертирования ключа key1 

30  gate LS key1,chan2

40 chan1  seize 1

50  advance 15

	60  release 1

70  transfer ,exit

80 chan2  seize 2

90  advance 17

100  release 2

200 exit  terminate 1

start 101

end




Формат записи блока LOGIC:

20 logic <X>  <A>

Во вспомогательном поле операций <X> задается  мнемоническое обозначение I — логический ключ под именем key1 инвертируется. В поле <A> задается номер или имя (в программе имя key1) логического ключа. Ссылка на имя ключа может происходить в любом месте программы. В поле <X> могут быть также заданы следующие мнемонические обозначения:

S — логический ключ устанавливается в состояние "включен";

R — логический ключ устанавливается в состояние "выключен".

Формат записи блока GATE для проверки логических ключей:

30 gate  <X>  <A>,<B>

Во вспомогательном поле <X> задается один из следующих логических операторов:

LS — логический ключ, заданный в поле <A>, включен;

LR — логический ключ, заданный в поле <A>, выключен.

В поле <A> задается номер или имя логического ключа, состояние котрого проверяется (в программе key1). В поле <B> задается номер блока, к которому переходит транзакт, если логический оператор вспомогательного поля <X> имеет значение "ложь" (в программе устройство с меткой chan2). Если значение логического оператора — "истина", транзакт переходит к следующему по номеру блоку (в устройство с меткой chan1 под номером 40). 
Задания к примерам 9, 11. 

· Для примера 9 изменить средние времена обработки: 60 и 90, 60 и 60, 120 и 60, 60 и 120, соответственно для первого и второго устройств. 
· Для примера 11 применить режим LS  блока GATE  для изменения порядка прохождения через устройства четных и нечетных транзактов. В исходной программе  применить режимы "S" и "R" для блока LOGIC. Написать программу, в которой четные транзакты обрабатываются в устройстве пятикратным циклом, а нечетные транзакты обрабатываются семикратным циклом. В программе изменить имя логического ключа, задать какой-либо числовой номер логического ключа. 

· Изучить и объяснить файлы стандартных отчетов рассмотренных программ.

